

TSPL 编程手册

Ver 1.01

厦门信远通科技有限公司

Xiamen XinYuanTong Technology Co., Ltd

地址：厦门市湖里区梧桐西路 1 号 411 室

电话：13063031816

邮箱：xyt.xm@aliyun.com

网址：<http://xmxyt.tech>

阅读规则

MPDF417

定义一个Micro PDF417。

语法:

MPDF417 x,y,rotate,[Wn,][Hn,][Cn],"content"

参数	描述
x	横向起始坐标（单位：点）
y	纵向起始坐标（单位：点）
rotate	旋转 0 : 不旋转 90 : 旋转90° 180: 旋转180° 270: 旋转270°
Wn	可选项；模块宽度（单位：点），默认1。
Hn	可选项；模块高度（单位：点），默认10。
Cn	可选项；列数。一旦设置列数，打印机将会根据内容自动调整行数。 0: 自动模式 1: 列数为1, 适合的行数为 11,14,17,20,24 和 28。 2: 列数为2, 适合的行数为 8,11,14,17,20,23 和 26。 3: 列数为3, 适合的行数为 6,8,10,12,15,20,26,32,38 和 44。 4: 列数为1, 适合的行数为 4,6,8,10,12,15,20,26,32,38 和 44。
"content"	条码内容

示例:

- 示例代码

```

SIZE 4,1
GAP 0,0
CLS
MPDF417 10,10,0, "ABCDEFGHijklmnopqrstuvwxyz0123456789 "
MPDF417 110,10,0,W2, "ABCDEFGHijklmnopqrstuvwxyz0123456789 "
MPDF417 210,10,0,W2,H3, "ABCDEFGHijklmnopqrstuvwxyz0123456789 "
MPDF417 310,10,0,W2,H3,C3, "ABCDEFGHijklmnopqrstuvwxyz0123456789 "
PRINT 1
  
```


- 打印输出



文档规则

条例	描述
[表示内容]	方括号里的内容为可选项，最大长度为2×1024字节
	ASCII 27, 状态查询指令的控制字符/实时指令
~	ASCII 126, 状态查询指令的控制字符/仅在打印机就绪时执行
Space	ASCII 32, 在命令行中忽略改字符
"	ASCII 32, 表达式开始与结束字符
CR,LF	ASCII 13, ASCII 10, 指示命令行结束
NULL	ASCII 0, 在表达式中支持
注意: <i>203DPI: 1mm = 8dots</i>	斜体表示注释


位置计算




注意:

1. 203DPI, 1mm=8dots; 300DPI, 1mm=11.8dots
2. 计算后只是用整数部分, 例如: 2mm = 23.6dots, 实际使用为23dots

标签起始位置(0,0)



DIRECTION 1



```
DIRECTION 1  
CLS  
TEXT 56,24,"3",0,1,1,"ABC"  
PRINT 1
```

设置和系统命令

SIZE

定义标签宽度与高度。

语法

SIZE m,n	英制系统 (英寸)
SIZE m mm,n mm	公制系统 (毫米)
SIZE m dot,n dot	点

参数	描述
M	标签宽度 (单位: 英寸/毫米/点)
N	标签高度 (单位: 英寸/毫米/点)

注意:


- 200 DPI: 1 mm = 8 dots
300 DPI: 1 mm = 12 dots
- 使用毫米和点为单位时, 参数与"mm","dot"之间必须有空格

示例:

- 示例代码

```
英寸:  
SIZE 3.5,3.00  
  
毫米:  
SIZE 100 mm,100 mm
```

- 图示



- 相关指令: GAP, BLINE

GAP

定义两张标签之间的缝宽。

语法:

GAP m,n	英制系统 (英寸)
GAP m mm,n mm	公制系统 (毫米)
GAP m dot,n dot	点

参数	描述
m	两个标签之间的距离
n	缝的偏移
0,0	连续纸

示例:

- 示例代码

- 正常缝标

英寸:
GAP 0.12,0

毫米:
GAP 3 mm,0 mm


连续:
GAP 0,0

- 特殊缝标

英寸:
GAP 0.30,0.10

毫米:
GAP 7.62 mm,2.54 mm

- 图示



- 相关指令：SIZE, BLINE

GAPDETECT

通过走纸测量标签纸面和缝的高度。测量值以用户输入的估计值为参考，如果测量结果与实际值冲突，则该指令不起作用。

语法：

GAPDETECT [x,y]

参数	描述
x	纸面高度（单位：点）
y	缝高度（单位：点）

注意：如果省略参数x,y，打印机将自动计算纸面和缝的高度。

- 相关指令：GAP,SIZE,BLINEDETECT,AUTODETECT

BLINEDETECT

通过走纸测量黑标纸纸张和黑标的高度。测量值以用户输入的估计值为参考，如果测量结果与实际值冲突，则该指令不起作用。

语法：

BLINEDETECT [x,y]

参数	描述
x	纸面高度（单位：点）
y	黑标高度（单位：点）

注意：如果省略参数x,y, 打印机将自动计算纸张和黑标的高度。

- 相关指令：GAP,SIZE,GAPDETECT,AUTODETECT

AUTODETECT

通过走纸测量黑标纸或标签纸张和黑标的高度。测量值以用户输入的估计值为参考，如果测量结果与实际值冲突，则该指令不起作用。

语法：

AUTODETECT [x,y]

参数	描述
x	纸面高度（单位：点）
y	黑标高度（单位：点）

注意：如果省略参数x,y, 打印机将自动计算纸张和黑标的高度。

- 相关指令：GAP,SIZE,GAPDETECT,BLINEDETECT

BLINE

用于设置黑标的高度以及用户自定义的每次定位额外走纸高度。

语法：

BLINE m,n	英寸
BLINE m mm,n mm	毫米
BLINE m dot,n dot	点

参数	描述
m	黑标高度 $0 \leq m \leq 5$ （英寸）， $0 \leq m \leq 127$ （毫米）
n	额外走纸高度 $0 \leq n \leq$ 标签高度
0,0	连续纸


示例：

- 示例代码

英寸：
BLINE 0.20,0.50

毫米：
BLINE 5.08 mm,12.7 mm

- 图示



- 相关指令：SIZE,GAP

OFFSET

定义每次标签定位时可选择的、额外的标签进纸高度。主要在撕纸或切纸模式下，用于调整标签停止位置，并在下次打印前回退响应的距离。

语法：

OFFSET m	英制系统 (英寸)
OFFSET m mm	公制系统 (毫米)
OFFSET m dot	点

参数	描述
m	偏移值 -1 ≤ m ≤ 1 (英寸)

注意：

- 不合适的偏移值可能引起卡纸


示例：

- 示例代码

```
英寸
OFFSET 0.5

毫米：
OFFSET 12.7 mm
```

- 图示



- 相关指令：SIZE,GAP,SET PEEL,SET CUTTER

SPEED

设置打印速度。

语法：

SPEED n

参数	描述
n	打印速度：英寸/秒

示例：

- 示例代码

```
SPEED 4
```

- 相关指令：DENSITY

DENSITY

设置打印浓度。

语法：

DENSITY n

参数	描述
n	0~15
	0：最淡
	15：最浓
	8：默认浓度

示例:

- 示例代码

```
DENSITY 6
```

DIRECTION 和 镜像

设定打印方向和镜像打印。

语法:

```
DIRECTION n,[m]
```


参数	描述
n	0 或 1
m	0: 正常打印 1: 镜像打印

示例:

- 示例代码

```
DIRECTION 0  
DIRECTION 0,1
```

- 图示



- 相关指令: REFERENCE

REFERENCE

设定标签参考点（起始点），参考点根据方向不同而不同。

语法:

REFERENCE x,y


参数	描述
x	横向起始坐标（单位：点）
y	纵向起始坐标（单位：点）

示例:

- 示例代码

```
REFERENT 10,10
```

- 图示



- 相关指令: REFERENCE

SHIFT

移动标签的横向和纵向位置。正数使标签往打印方向的相反方向移动，负数使标签往打印方向移动。

语法:

SHIFT [x,y]

参数	描述
x	可选项。横向移动位置
y	纵向移动位置


示例:

- 示例代码

```

SIZE 4,2.5
GAP 2 mm,0
DIRECTION 0
SHIFT 36
OFFSET 0
CLS
TEXT 400,200,"3",0,1,1,"DIRECTION 0"
TEXT 400,250,"3",0,1,1,"SHIFT 36"
BOX 10,0,780,490,8
PRINT 3,1
    
```

- 图示



- 相关指令: OFFSET,REFERENCE

COUNTRY

选择国际字符集。

语法:

COUNTRY n

参数	描述
n	001: USA 002: Canadian-French 003: Spanish (Latin America) 031: Dutch 032: Belgian 033: French (France) 034: Spanish (Spain) 036: Hungarian 038: Yugoslavian 039: Italian 041: Switzerland 042: Slovak 044: United Kingdom 045: Danish 046: Swedish 047: Norwegian 048: Polish 049: German 055: Brazil 061: English (International) 351: Portuguese 358: Finnish

示例:

```
COUNTRY 001
```

- 相关指令: CODEPAGE, ~!!

CODEPAGE

设置代码页及国际字符集

语法:

```
CODEPAGE n
```

参数	描述
n	代码页

7位codepage		8位codepage		Windows		ISO	
n	名称	n	名称	n	名称	n	名称
USA	USA	437	United States	1250	Central Europe	8859-1	Latin 1
BRI	British	737	Greek	1251	Cyrillic	8859-2	Latin 2
GER	German	850	Multilingual	1252	Latin I	8859-3	Latin 3

FRE	French	851	Greek 1	1253	Greek	8859-4	Baltic
DAN	Danish	852	Slavic	1254	Turkish	8859-5	Cyrillic
ITA	Italian	855	Cyrillic	1255	Hebrew	8859-6	Arabic
SPA	Spanish	857	Turkish	1256	Arabic	8859-7	Greek
SWE	Swedish	860	Portuguese	1257	Baltic	8859-8	Hebrew
SWI	Swiss	861	Icelandic	1258	Vietnam	8859-9	Turkish
		862	Hebrew	932	Japanese Shift-JIS	8859-10	Latin 6
		863	Canadian/ French	936	GBK	8859-15	Latin 9
		864	Arabic	949	Korean		
		865	Nordic	950	BIG5		
		866	Russian	UTF-8	UTF 8		
		869	Greek 2				

示例:

- 示例代码
- 相关指令: CODEPAGE, ~!!

CLS

清除画布缓存。

语法:

CLS

参数	描述
无	N/A

- 注意: 该命令必须在SIZE命令之后

示例:

```
CLS
```

- 相关指令: SIZE,GAP,BLINE

FEED

打印机走纸指定的长度。

语法:

FEED n

参数	描述
n	单位: 点, $1 \leq n \leq 99999$

注:

200 DPI: 1mm = 8点

300 DPI: 1mm = 12点

示例:

```
FEED 40
```

- 相关指令: BACKFEED,SIZE,GAP,BLINE,HOME,FORMFEED

BACKFEED

打印机回退指定的长度。

语法:

BACKFEED n

参数	描述
n	单位: 点, $1 \leq n \leq 99999$

注:

200 DPI: 1mm = 8点

300 DPI: 1mm = 12点

- 注意: 回退不合适的长度可能引起卡纸

示例:

```
BACKFEED 40
```

- 相关指令: FEED,SIZE,GAP,BLINE,HOME,FORMFEED

FORMFEED

定位到下一张标签。

语法:

FORMFEED

参数	描述
无	N/A

- 注意: 该指令必须放在SIZE之后

示例:

- 示例代码

```
SIZE 4,2.5  
GAP 2 mm,0  
DIRECTION 1  
FORMFEED  
CLS  
TEXT 25,25,"3",0,1,1,"FORMFEED COMMAND TEST"  
PRINT 1,1
```

- 打印输出



- 相关指令: *FEED,SIZE,GAP,BLINE,HOME,BACKFEED*

HOME

走纸直到传感器侦测到一个起始位置。在用此命令之前需要先设置标签的大小和缝。

语法:**HOME**

参数	描述
无	N/A

示例:

• 示例代码

```
SIZE 4,2.5
GAP 2 mm,0
SET COUNTER @0 +1
@0="000001"
HOME
CLS
BOX 1,1,360,65,12
TEXT 25,25, "3",0,1,1, "HOME COMMAND TEST"
TEXT 25,80, "3",0,1,1,@0
PRINT 3,1
```

• 相关指令: FEED,SIZE,GAP,BLINE,FORMFEED

PRINT

打印已缓存的标签。

语法:**PRINT m[,n]**


参数	描述
m	打印标签的组数
n	每组标签打印的数量

示例:

• 示例代码

```
SIZE 50 mm,25 mm
GAP 3 mm,0
DIRECTION 1
SET COUNTER @1 1
@1="0001"
CLS
TEXT 10,10, "3",0,1,1,@1
PRINT 3,2
```

• 图示:



SOUND

控制蜂鸣器的发声频率。总共有10个等级，间隔时间由"interval"参数控制。

语法：

SOUND level,interval

参数	描述
level	频率等级：0~9
interval	间隔时间：1~4095

示例：

- 示例代码

```
SOUND 5,200
SOUND 3,200
SOUND 3,200
SOUND 4,200
SOUND 2,200
SOUND 2,200
SOUND 1,200
SOUND 2,200
SOUND 3,200
SOUND 4,200
SOUND 5,200
```

CUT

立即切纸，不回退标签。

语法：

CUT

参数	描述
无	N/A

示例：

- 示例代码

```
SIZE 3,3
GAP 0,0
CLS
BOX 0,0,866,866,5
TEXT 100,100, "5",0,1,1, "FEED & CUT"
TEXT 100,200, "5",0,1,1, "300 DPI"
PRINT 1,1
FEED 260
CUT
```

- 相关指令：SET CUTTER,SET BACK,SET PARTIAL_CUTTER

LIMITFEED

如果传感器设置异常，打印机将不能正常定位到标签缝。该指令用于设定最大的侦测长度，打印机走过此最大长度仍没有定位到一个缝时启动LED或蜂鸣器警报。

语法：

LIMITFEED n[,minpaper,maxgap]	英制系统
LIMITFEED n mm[,minpaper mm,maxgap mm]	公制系统
LIMITFEED n dot[,minpaper dot,maxgap dot]	点

描述	描述
n	最大侦测长度
minpaper	标签的最小长度
maxgap	缝的最大长度

示例:

- 示例代码

```
英制系统（英寸）：  
LIMITFEED 12
```

SELFTEST

打印测试页。

语法:**SELFTEST [page]**

参数	描述
page	省略：完整的打印机信息 PATTERN：机芯测试图形 ETHERNET：以太网设置信息 WLAN：Wi-Fi设置信息 RS232：串口设置信息 SYSTEM：打印机设置信息 Z：打印语言信息 BT：蓝牙信息

示例:

- 示例代码

```
SELFTEST  
SELFTEST PATTERN  
SELFTEST SYSTEM  
SELFTEST PRINTER  
SELFTEST Z
```

- 打印输出

EOJ

通知打印机等待EOJ之前的指令处理完后再处理之后的命令。

语法:**EOJ**

参数	描述
无	N/A

示例:

- 示例代码

```

SIZE 4,0.2
GAP 0,0
DIRECTION 1
CLS
TEXT 10,10,"3",0,1,1,"Two labels are printed without stop."
PRINT 1
PRINT 1
SIZE 4,0.2
GAP 0,0
CLS
TEXT 10,10,"3",0,1,1,"Printer stops before next printing."
PRINT 1
EOJ
PRINT 1
    
```

- 打印输出

进纸方向



```

Printer stops before next printing.
Printer stops before next printing.
Two labels are printed without stop.
Two labels are printed without stop.
    
```

} 不停顿等待

DELAY

等待指定的时间后再执行后续的命令。

语法:

DELAY ms

参数	描述
ms	等待周期 (单位: 毫秒, 1000毫秒 = 1秒)

示例:

- 示例代码


```

SIZE 4,0.7
GAP 0,0
DIRECTION 1
CLS
TEXT 10,10,"3",0,1,1,"The delay time between two labels is 3 seconds."
TEXT 10,60,"3",0,1,1,"Now second:" +@SECOND
PRINT 1
DELAY 3000
PRINT 1
  
```

- 打印输出

进纸方向



The delay time between two labels is 3 seconds.

Now second: 9

The delay time between two labels is 3 seconds.

Now second: 6

DISPLAY

在LCD面板上显示打印缓存的图形。

语法:

DISPLAY IMAGE/OFF/CLS/...

参数	说明
IMAGE	显示打印缓存内容
OFF	关闭此功能
CLS	清除LCD

- 注意: 预留指令, 暂不支持

INITIALPRINTER

恢复打印机到默认设定。

语法:

INITIALPRINTER

参数	描述
无	N/A

示例:

INITIALPRINTER

MENU

根据打印机内置数据库定制用户菜单。

语法:

MENU title, list, selected

参数	描述
title\$	显示菜单标题
list\$	显示列表项, 以CRLF分隔
selected	必须是获取选项结果的一个变量

- 注意: 预留指令, 暂不支持

标签格式化指令

BAR

绘制线条。

语法:

BAR x,y,width,height


参数	描述
x	左上角的 x 坐标 (单位: 点)
y	左上角的 y 坐标 (单位: 点)
width	线条宽度 (单位: 点)
height	线条高度 (单位: 点)

示例:

- 示例代码

```
SIZE 50 mm,25 mm
GAP 3 mm,0
DIRECTION 1
CLS
BAR 80,80,300,100
PRINT 1,1
```

- 打印结果



- 相关指令: BOX

BARCODE

打印一维条码。

语法:

```
**BARCODE X,Y,"code type",height,human readable,rotation,narrow,wide,[alignment,]"content"
```

参数	描述
x	x 坐标
y	y 坐标
code type	条码类型, (详见[表BAR1D])
height	条码高度(单位: 点)
human readable	可读字符
rotation	0 : 不旋转 90 : 顺时针旋转90° 180: 顺时针旋转180° 270: 顺时针旋转270°
narrow	窄条宽度(单位: 点)
wide	宽条宽度(单位: 点), (详见[表BAR1D]"窄条:宽条")
alignment	条码对齐方式 0: 默认(左对齐) 1: 左对齐 2: 居中 3: 右对齐
content	条码内容, (详见[表BARC])

表BAR1D

条码类型	描述	窄条:宽条					最大数据长度
		1:1	1:2	1:3	2:5	3:7	
128	Code 128, 自动编码	10x	-	-	-	-	
128M	Code 128, 手动编码 控制字符见[表Code128M]	10x	-	-	-	-	

EAN128	EAN128, 自动编码	10x	-	-	-	-	
EAN128M	EAN128, 手动编码	10x	-	-	-	-	
25	交叉25码	-	10x	10x	5x	-	偶数
25C	交叉25码, 带校验	-	10x	10x	5x	-	奇数
25S	标准25码	-	10x	10x	5x		
25I	工业25码	-	10x	10x	5x		
39	Code 39, 自动切换标准模式与全ASCII模式	-	10x	10x	5x	-	
39C	Code 39, 带校验	-	10x	10x	5x	-	
93	Code 93	-	-	10x	-	-	
EAN13	EAN 13	8x	-	-	-	-	12
EAN13+2	EAN 13及2位附加码	8x	-	-	-	-	14
EAN13+5	EAN 13及5位附加码	8x	-	-	-	-	17
EAN8	EAN 8	8x	-	-	-	-	7
EAN8+2	EAN 8及2位附加码	8x	-	-	-	-	9
EAN8+5	EAN 8及5位附加码	8x	-	-	-	-	12
CODA	Codabar	-	10x	10x	5x	-	
POST	Postnet	1x	-	-	-	-	5,9,11
UPCA	UPC-A	8x	-	-	-	-	11
UPCA+2	UPC-A及2位附加码	8x	-	-	-	-	13
UPCA5	UPC-A及5位附加码	8x	-	-	-	-	16
UPCE	UPC-E	8x	-	-	-	-	6
UPCE2	UPC-E及2位附加码	8x	-	-	-	-	8
UPCE5	UPC-E及5位附加码	8x	-	-	-	-	11
MSI	MSI	-	-	10x	-	-	
MSIC	MSI, 带校验	-	-	10x	-	-	
PLESSEY	PLESSEY	-	-	10x	-	-	
CPOST	中国邮政码	-	-	-	-	1x	
ITF14	ITF14	-	10x	10x	5x	-	13
EAN14	EAN14	8x	-	-	-	-	13
11	Code 11	-	10x	10x	5x		
TELEPEN	Telepen		V	V	V		
TELEPENN	Telepen 数字		V	V	V		

PLANET	Planet	V					
CODE49	Code 49	V					
DPI	德国邮政 Identcode		V	V	V		11
DPL	德国邮政 Leitcode		V	V	V		13
LOGMARS	Code 39的一种特殊用法		V	V	V		

- 注意：以上表中条码支持参考具体机型

表Code128M

控制码	A	B	C
096	FNC3	FNC3	无
097	FNC2	FNC2	无
098	SHIFT	SHIFT	无
099	CODE C	CODE C	无
100	CODE B	FNC4	CODE B
101	FNC4	CODE A	CODE A
102	FNC1	FNC1	FNC1
103	Start(CODE A)		
104	Start(CODE B)		
105	Start(CODE C)		

- 注意：使用一个"!"字符加一个3位控制字符组成控制码，如未设置，默认为 CODE B.

表BARC

条码类型	字符集	最大长度
128	参考 CODE128 字符集	-
128M	参考 CODE128 字符集	-
EAN128	参考 CODE128 字符集	-
EAN128M	参考 CODE128 字符集	-
25	0123456789	长度为偶数
25C	0123456789	长度为奇数
25S	0123456789	
25I	0123456789	
39 I	0123456789[Space]ABCDEFGHIJKLMNOPQRSTUVWXYZ-.\$/+%	-

39 I 全ASCII	0123456789[Space]ABCDEFGHIJKLMNOPQRSTUVWXYZ-.\$/+%&'()*+,-./:;<>? @[\\]^_`abcdefghijklmnopqrstuvwxyz{ }~	-
93	0123456789[Space]ABCDEFGHIJKLMNOPQRSTUVWXYZ-.\$/+%&'()*+,-./:;<>? @[\\]^_`abcdefghijklmnopqrstuvwxyz{ }~	-
EAN13	0123456789	12
EAN13+2	0123456789	14
EAN13+5	0123456789	17
EAN8	0123456789	7
EAN8+2	0123456789	9
EAN8+5	0123456789	12
CODA	0123456789-\$./+	-
POST	0123456789	5,9,11
UPCA	0123456789	11
UPCA+2	0123456789	13
UPCA+5	0123456789	16
UPCE	0123456789	6
UPCE+2	0123456789	8
UPCE+5	0123456789	11
MSI	0123456789	-
MSIC	0123456789	-
PLESSEY	0123456789	-
CPOST	0123456789	-
ITF14	0123456789	13
EAN14	0123456789	13
11	0123456789-	-
TELEPEN	ASCII 0 到 127	30
TELEPENN	0123456789	60
PLANET	0123456789	38
CODE49	ASCII 0 到 127	81
DPI	0123456789	11
DPL	0123456789	13
LOGMARS	0123456789[Space]ABCDEFGHIJKLMNOPQRSTUVWXYZ-.\$/+%	-

CODE 128 字符集

Value	128A	128B	128C	Value	128A	128B	128C	Value	128A	128B	128C
0	space	space	00	36	D	D	36	72	BS	h	72
1	!	!	01	37	E	E	37	73	HT	i	73
2	"	"	02	38	F	F	38	74	LF	j	74
3	#	#	03	39	G	G	39	75	VT	k	75
4	\$	\$	04	40	H	H	40	76	FF	l	76
5	%	%	05	41	I	I	41	77	CR	m	77
6	&	&	06	42	J	J	42	78	SO	n	78
7	'	'	07	43	K	K	43	79	SI	o	79
8	((08	44	L	L	44	80	DLE	p	80
9))	09	45	M	M	45	81	DC1	q	81
10	*	*	10	46	N	N	46	82	DC2	r	82
11	+	+	11	47	O	O	47	83	DC3	s	83
12	,	,	12	48	P	P	48	84	DC4	t	84
13	-	-	13	49	Q	Q	49	85	NAK	u	85
14	.	.	14	50	R	R	50	86	SYN	v	86
15	/	/	15	51	S	S	51	87	ETB	w	87
16	0	0	16	52	T	T	52	88	CAN	x	88
17	1	1	17	53	U	U	53	89	EM	y	89
18	2	2	18	54	V	V	54	90	SUB	z	90
19	3	3	19	55	W	W	55	91	ESC	{	91
20	4	4	20	56	X	X	56	92	FS		
21	5	5	21	57	Y	Y	57	93	GS	}	93
22	6	6	22	58	Z	Z	58	94	RS	~	94
23	7	7	23	59	[[59	95	US	DEL	95
24	8	8	24	60	\	\	60	96	FNC 3	FNC 3	96
25	9	9	25	61]]	61	97	FNC 2	FNC 2	97
26	:	:	26	62	^	^	62	98	Shift B	Shift A	98
27	;	;	27	63	_	_	63	99	Code C	Code C	99
28	<	<	28	64	NUL	`	64	100	Code B	FNC 4	Code B
29	=	=	29	65	SOH	a	65	101	FNC 4	Code A	Code A
30	>	>	30	66	STX	b	66	102	FNC 1	FNC 1	FNC 1

Value	128A	128B	128C	Value	128A	128B	128C	Value	128A	128B	128C
31	?	?	31	67	ETX	c	67	103	Start-	Code-	A
32	@	@	32	68	EOT	d	68	104	Start-	Code-	B
33	A	A	33	69	ENQ	e	69	105	Start-	Code-	C
34	B	B	34	70	ACK	f	70				
35	C	C	35	71	BEL	g	71				

- 注意: *Start-Code-A*, *Start-Code-B*, *Start-Code-C*为合并行.

示例:

- 示例代码

```

SIZE 4,1
GAP 0,0
DIRECTION 1
CLS
TEXT 10,10, "2",0,1,1, "Human readable alignment"
BARCODE 10,50, "128",100,1,0,2,2,"left"
BARCODE 310,50, "128",100,2,0,2,2,"center"
BARCODE 610,50, "128",100,3,0,2,2,"right"
PRINT 1

```

- 打印输出

Human readable alignment



left



center



right


```

SIZE 4,1
GAP 0,0
DIRECTION 1
CLS
TEXT 10,10, "2",0,1,1, "Code 128, switch code subset
automatically. "
BARCODE 10,50, "128",100,1,0,2,2, "123456abcd123456"
PRINT 1

```

- 打印输出

Code 128, switch code subset automatically.



123456abcd123456

```
SIZE 4,1  
GAP 0,0  
DIRECTION 1  
CLS  
TEXT 10,10, "2",0,1,1, "Code 128, switch code subset manually."  
BARCODE 10,50, "128M",100,1,0,2,2, "!104!096ABCD!101EFGH"  
PRINT 1
```

- 打印输出

Code 128, switch code subset manually.



ABCDEFGH

```
SIZE 4,1  
GAP 0,0  
DIRECTION 1  
CLS  
TEXT 10,10, "2",0,1,1, "TELEPEN"  
BARCODE 10,50, "TELEPEN",100,1,0,2,6, "abcd1234ABCD"  
PRINT 1
```

- 打印输出

TELEPEN



abcd1234ABCD

```

SIZE 4,4
GAP 0,0
DIRECTION 1
CLS
TEXT 400,26, "2",0,1,1,2, "TELEPEN Number"
BARCODE 400,50, "TELEPEN",60,2,0,2,6,2, "1234567890"
TEXT 400,136, "2",0,1,1,2, "Code 11"
BARCODE 400,160, "11",60,2,0,2,6,2, "1234567890"
TEXT 400,246, "2",0,1,1,2, "PLANET"
BARCODE 400,270, "PLANET",60,2,0,2,2,2, "12345678901"
TEXT 400,356, "2",0,1,1,2, "Deutsche Post Identcode."
BARCODE 400,380, "DPI",60,2,0,2,6,2, "12345678901"
TEXT 400,466, "2",0,1,1,2, "Deutsche Post Leitcode. "
BARCODE 400,490, "DPL",60,2,0,2,6,2, "123456789012"
PRINT 1
  
```

- 打印输出



TLC39

绘制 TLC39 码，TLC39 为 MicroPDF7 与 Code39 的组合码。

语法：

```
TLC39 x,y,rotation,[height],[narrow],[wide],[cellwidth],[cellheight,]"ECI number,Serial number & additional data"
```

参数	描述
x	x 坐标
y	y 坐标
rotation	0 : 不旋转 90 : 顺时针旋转90° 180: 顺时针旋转180° 270: 顺时针旋转270°
height	Code 39 部分高度 (单位: 点)
narrow	Code 39 窄条宽度 (单位: 点, 默认: 2)
wide	Code 39 宽条宽度 (单位: 点, 默认: 4)
cellwidth	MicroPDF417 单元宽度 (单位: 点, 默认: 2)
cellheight	MicroPDF417 单元高度 (单位: 点, 默认: 4)
ECI number	用于产生 Code 39, 必须是6个数字
Serial number & additional data	字母, Micro-PDF417内容

- 注意: "ECI number, Serial number & additional data"的双引号(")是必须的

示例:

```

SIZE 4,1.2
GAP 0,0
DIRECTION 1
CLS
TEXT 10,10,"3",0,1,1,"TLC39 code"
TLC39 10,50,0,"123456,SN00000001,00601,01501"
TLC39 310,50,0,80,3,6,3,4,"123456,SN00000001,00601,01501"
PRINT 1
    
```

PDF417

绘制PDF417码

语法:

```
PDF417 x,y,width,height,rotate[,option],"content"
```

参数	描述
x	起始点 x 坐标 (单位: 点)
y	起始点 y 坐标 (单位: 点)
width	预期宽度 (单位: 点)
height	预期高度 (单位: 点)
rotate	旋转方向 0: 不旋转 90: 旋转90° 180: 旋转180° 270: 旋转270°
option	可选项。 P 数据压缩方式 0: 自动编码 1: 二进制模式 E 纠错等级 (范围: 0~8) M 图形在条码区域打印位置 0: 靠左 1: 居中 Ux,y,c 可读字符打印 x: x 坐标 y: y 坐标 c: 一行最大打印字符个数 W 模块宽度 (单位: 点, 范围: 2~9) H 条码高度 (单位: 点, 范围: 4~99) R 最大行数 C 最大列数 T 截断 0: 不截断 1: 截断 Lm 数据长度, $1 \leq m \leq 2048$ (不包括")
content	PDF417 数据

示例:

- 示例代码1

```

SIZE 4,1
GAP 0,0
DIRECTION 1
REM *****WITHOUR OPTIONS*****
CLS
PDF417 50,50,400,200,0, "Without Options"
PRINT 1,1
  
```

- 打印输出



- 示例代码2

```

SIZE 4,1.5
GAP 0,0
DIRECTION 1
REM *****OPTION:E3*****
CLS
PDF417 50,50,400,200,0,E3, "Error correction level:3"
PRINT 1,1
REM *****OPTION:E4*****
CLS
PDF417 50,50,400,200,0,E4, "Error correction level:4"
PRINT 1,1

```

- 打印输出



- 示例代码3

```

SIZE 4,1.5
GAP 0,0
DIRECTION 1
REM *****OPTION:E4 W4*****
CLS
PDF417 50,50,600,600,0,E4,W4, "Error correction level:4 module
width 4 dots"
PRINT 1,1
REM *****OPTION:E4 W4 H4*****
CLS
PDF417 50,50,600,600,0,E4,W4,H4, "Error correction level:4
module width 4 dots bar height 4 dots"
PRINT 1,1

```

- 打印输出



- 示例代码4

```

SIZE 4,1.5
GAP 0,0
DIRECTION 1
REM *****OPTION:E4 W4 H4 R40 C4 T1*****
CLS
PDF417 50,50,800,800,0,E4,W4,H4,R40,C4,T1, "Error correction
level:4
Module Width 4 dots
Bar Height 4 dots
Maximum Number of Rows:5 Rows
Maximum number of columns:90 Cols
Truncation:1"
PRINT 1,1
  
```

- 打印输出



- 示例代码5

```

SIZE 4,2.5
GAP 0,0
DIRECTION 1
REM *****OPTION:P1 E4 M1
U50,300,50,W4,H4,R60,C4,T0,L297*****
CLS
PDF417 0,50,576,600,0,P1,E4,M1,U50,300,50,W4,H4,R60,C4,T0,L297,Data compression method: P1
Error correction level: E4
Center pattern in barcode area: M1
Human Readable: Yes: U50,300,50
Module Width 4 dots: W4
Bar Height 4 dots: H4
Maximum Number of Rows: 60 Rows: R60
Maximum number of columns: 4 Cols: C4
Truncation:1: T0
Expression length:297: L297
PRINT 1,1
  
```

- 打印输出



MPDF417

绘制 Micro PDF417 码

语法:

MPDF417 x,y,rotate,[Wn,][Hn,][Cn,]"Content"

参数	描述
x	水平方向起始位置 (单位: 点)
y	垂直方向起始位置 (单位: 点)
rotate	旋转方向 0: 不旋转 90: 旋转90° 180: 旋转180° 270: 旋转270°
Wn	可选项。模块宽度 (单位: 点, 默认: 1)
Hn	可选项。模块高度 (单位: 点, 默认: 10)
Cn	可选项。列数。一旦设定列数, 打印机将自动计算行数
Content	Micro PDF417 编码数据

示例:

- 示例代码

```

SIZE 4,1
GAP 0,0
CLS
MPDF417 10,10,0, "ABCDEFGHJKLMNOPQRSTUVWXYZ0123456789 "
MPDF417 110,10,0,W2, "ABCDEFGHJKLMNOPQRSTUVWXYZ0123456789 "
MPDF417 210,10,0,W2,H3, "ABCDEFGHJKLMNOPQRSTUVWXYZ0123456789 "
MPDF417 310,10,0,W2,H3,C3, "ABCDEFGHJKLMNOPQRSTUVWXYZ0123456789 "
PRINT 1
    
```

- 打印输出



QR CODE

打印 QR 码。

语法：

```
QR CODE x,y,ECC level,cell width,mode,rotate,[justification,]model,],[mask,]area,]"content"
```


参数	描述
x	左上角的 x 坐标
y	左上角的 y 坐标
ECC level	纠错等级 L: 7% M: 15% Q: 25% H: 30%
cell width	模块宽度: 1~10
mode	编码模式 A: 自动 M: 手动
rotate	旋转 0: 不旋转 90: 旋转90° 180: 旋转180° 270: 旋转270°
justification	对齐方式 (范围: J1~J9)
model	M1: 默认, 原始版本 (不支持) M2: 加强版
area	最大条码区域 (X点: 例如: X100)
content	QR 码编码数据 可编码字符集: 1) 数字 (0~9) 2) 字母数字 数字 0~9 大写字母 A~Z 9个其它字符: 空格 \$ % * + - . / : 3) 8位字节数据 JIS 8位字符集 (Latin 和 Kana) 4) 日本汉字 Shift-JIS: 0x8140 ~ 0x9FFC, 0xE040 ~ 0xEAA4

一个QR码最大能容纳的字符数

	M1	M2
数字	1167	7089
字母数字	707	4296
8位数据	486	2953
Kanji	299	1817

- * 如果第一个字符为“A”，则紧随其后的数据为“字母数字”
- * 如果第一个字符为“N”，则紧随其后的数据为“数字”
- * 如果第一个字符为“B”，则紧随其后的数据为“8位字节数据”
- * 如果第一个字符为“K”，则紧随其后的数据为“日本汉字”
- * 如果数据中有“!”，并且跟着字母“N”、“A”、“B”、“K”中的一个，则切换为指定的字符集

手动模式:

```
A: QRCODE 100,10,L,7,M,0,M1,S1,"ATHE FIRMWARE HAS BEEN UPDATED"
N: QRCODE 100,10,M,7,M,0,M1,S2,"N123456"
N->A: QRCODE 100,10,Q,7,M,0,M1,S3,"N123456!ATHE FIRMWARE HAS BEENUPDATED"
B: QRCODE 100,10,H,7,M,0,M1,S3,"B0012Product name"
K: QRCODE 100,10,M,7,M,0,M1,S3,"Kおガ"
```

自动模式:

```
QRCODE 100,10,M,7,A,0,"THE FIRMWARE HAS BEEN UPDATED"
```

示例:

• 示例代码1

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,4,A,0,"ABCabc123"  
QRCODE 160,160,H,4,A,0,"123ABCabc"  
QRCODE 310,310,M,4,A,0,M2,"印表機 ABCabc123"  
PRINT 1,1
```

• 打印输出



• 示例代码2

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,4,A,0,"ABC<Enter>  
abc<Enter>  
123 "  
QRCODE 160,160,H,4,A,0,"123<Enter>  
ABC<Enter>  
abc"  
QRCODE 310,310,H,4,A,0,"印表機<Enter>  
ABC<Enter>  
abc<Enter>  
123"  
PRINT 1,1
```

• 打印输出



- 示例代码3

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,4,A,0,"ABCabc123"+STR$(1234)  
QRCODE 160,160,H,4,A,0,"123ABCabc"+"1234"  
QRCODE 310,310,H,4,A,0,"印表機  
ABCabc123"+"1234"+"abcd"  
PRINT 1,1
```

- 打印输出



• 示例代码4

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,4,A,0,"ABC\["]abc\["]123"  
QRCODE 160,160,H,4,A,0,"123\["]ABC\["]abc"  
QRCODE 310,310,H,4,A,0,"\["]印表機\["]ABCabc123"  
PRINT 1,1
```

• 打印输出



• 示例代码5

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,4,M,0,"AABC!B0003abc!N123"  
QRCODE 160,160,H,4,M,0,"N123!AABC!B0003abc"  
QRCODE 310,310,H,4,M,0,"K 印表機!AABC!B0006abc123"  
PRINT 1,1
```

- 打印输出



- 示例代码6

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,4,M,0,"AABC!B0007<Enter>  
abc<Enter>  
!N123"  
QRCODE 160,160,H,4,M,0,"N123!B0002<Enter>  
!AABC!B0005<Enter>  
abc"  
QRCODE 310,310,H,4,M,0,"K 印表機!B0002<Enter>  
!AABC!B0010<Enter>  
abc<Enter>  
123"  
PRINT 1,1
```

- 打印输出



• 示例代码7

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,4,M,0,"AABC!B0006abc123!N"+STR$(1234)  
QRCODE 160,160,H,4,M,0,"N123!AABC!B0007abc"+"1234"  
QRCODE 310,310,H,4,M,0,"K 印表機!AABC!B0014abc123"+"1234"+"abcd"  
PRINT 1,1
```

• 打印输出



• 示例代码8

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,4,M,0,"AABC!B0005\[ ]abc\[ ]!N123"  
QRCODE 160,160,H,4,M,0,"N123!B0001\[ ]!AABC!B0004\[ ]abc"  
QRCODE 310,310,H,4,M,0,"B0001\[ ]!K 印表機!B0010\[ ]ABCabc123"  
PRINT 1,1
```

- 打印输出



- 示例代码9

```
SIZE 3,3  
GAP 0,0  
DIRECTION 1  
CLS  
QRCODE 10,10,H,7,M,0,M2,S7,"Aabcd"  
QRCODE 170,170,H,4,M,0,M2,"B0008繁體中文"  
QRCODE 300,300,L,8,M,0,M2,"B0026http://www.tscprinters.com"  
PRINT 1,1
```


- 打印输出



• 示例代码10

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
BAR 60,120,200,1  
BAR 160,20,1,200  
QRCODE 160,120,H,10,A,0,X100,J5,"123456789"  
PRINT 1,1
```

• 打印输出



绘制 RSS 条码

语法:

```

RSS x,y,"sym",rotate,pixMult,sepHt,"content"
RSS x,y,"RSSEXP",rotate,pixMult,sepHt,segWidth,"content"
RSS x,y,"UCC128CCA",rotate,pixMult,sepHt,linHeight,"content"
RSS x,y,"UCC128CCC",rotate,pixMult,sepHt,linHeight,"content"
  
```

参数	描述	
x	x 坐标	
y	y 坐标	
sym	RSS14	RSS14
	RSS14T	RSS14 Truncated
	RSS14S	RSS14 Stacked
	RSS14S0	RSS14 Stacked Omnidirectional
	RSSLIM	RSS14 Limited
	RSSEXP	RSS Expanded
	UPCA	UPC-A
	UPCE	UPC-E
	EAN13	EAN-13
	EAN8	EAN-8
	UCC128CCA	UCC/EAN-128 & CC-A/B
	UCC128CCC	UCC/EAN-128 & CC-C
	rotate	旋转方向 (0,90,180 和 270)
	pixMult	模块宽度 (单位: 点, 范围: 1 ~ 10)
条码高度计算:		
RSS14		33 × pixMult
RSS14T		13 × pixMult
RSS14S		13 × pixMult
RSS14S0		33 × pixMult
RSSLIM		13 × pixMult
RSSEXP		33 × pixMult
EAN8		60 × pixMult
EAN13		74 × pixMult
UPCA		74 × pixMult
UPCE		74 × pixMult
sepHt		分隔层高度 (1或2, 实际高度 pixMult × sepHt)
setWidth		RSSEXP 段宽度 (偶数: 2~22)
linHeight	UCC/EAN-128高度 (单位: 点, 范围: 1~500)	
content	条码内容	
	UPCE数据必须按如下格式:	
	* 00abc0000hij = abhijc, 其中 c = 0 - 2	
	* 00abc00000ij = abcij3	
	* 00abcd00000j = abcdj4	
* 00abcde0000j = abcdej, 其中 j = 5 - 9		

示例:

- 示例代码1

```
SIZE 100 mm,100 mm
GAP 0,0
DIRECTION 1
CLS
RSS 300,300, "RSS14",0,2,2, "1234567890|ABCDEFGF"
RSS 300,300,"RSS14T",90,2,2, "1234567890|ABCDEFGF"
RSS 300,300,"RSS14S",180,2,2, "1234567890|ABCDEFGF"
RSS 300,300, "RSS14SO",270,2,2, "1234567890|ABCDEFGF"
PRINT 1,1
```

- 打印输出
- 示例代码2

```
SIZE 100 mm,100 mm
GAP 0,0
DIRECTION 1
CLS
RSS 300,300, "RSSLIM",0,2,2, "1234567890|ABCDEFGF"
RSS 300,300, "RSSEXP",90,2,2,22, "1234567890|ABCDEFGF"
RSS 300,300, "UPCA",180,2,2, "1234567890|ABCDEFGF"
RSS 300,300, "UPCE",270,2,2, "000|ABCDEFGF"
PRINT 1,1
```

- 打印输出
- 示例代码3

```
SIZE 100 mm,100 mm
GAP 0,0
DIRECTION 1
CLS
RSS 300,300,"EAN13",0,2,2,"123456789012|ABCDEFGF"
RSS 300,300,"EAN8",90,2,2,"1234567|ABCDEFGF"
RSS 300,300,"UCC128CCA",180,2,2,25,"1234567890|ABCDEFGF"
RSS 300,300,"UCC128CCC",270,2,2,25,"1234567890|ABCDEFGF"
PRINT 1,1
```

- 打印输出
- 示例代码4

```
SIZE 100 mm, 100 mm
GAP 0,0
DIRECTION 1
CLS
RSS 300,10, "RSSEXP",90,2,2,12,
"81101061414112345628911012012120850100480002140256100
48000310123191000"
PRINT 1
```

- 打印输出
- 示例代码5

```
SIZE 4,1
GAP 0,0
DIRECTION 1
CLS
REM UPCE Rule 1: 00abc0000hij = abhijc, where c = 0-2
RSS 10,10,"UPCE",0,2,2,"001200000456|ABCDEFGF"
RSS 210,10,"UPCE",0,2,2,"001210000456|ABCDEFGF"
RSS 410,10,"UPCE",0,2,2,"001220000456|ABCDEFGF"
PRINT 1
```

- 打印输出
- 示例代码6

```
SIZE 4,1
GAP 0,0
DIRECTION 1
CLS
REM UPCE Rule 2: 00abc00000ij = abcij3
RSS 10,10,"UPCE",0,2,2,"001230000045|ABCDEFGF"
PRINT 1
SIZE 4,1
CLS
REM UPCE Rule 3: 00abcd00000j = abcdj4
RSS 10,10,"UPCE",0,2,2,"001234000005|ABCDEFGF"
PRINT 1
```

- 打印输出
- 示例代码7

```
SIZE 4,1
GAP 0,0
DIRECTION 1
CLS
REM UPCE Rule 4: 00abcde0000j = abcdej where j = 5-9
RSS 10,10,"UPCE",0,2,2,"001234500005|ABCDEFGF"
RSS 160,10,"UPCE",0,2,2,"001234500006|ABCDEFGF"
RSS 310,10,"UPCE",0,2,2,"001234500007|ABCDEFGF"
RSS 460,10,"UPCE",0,2,2,"001234500008|ABCDEFGF"
RSS 610,10,"UPCE",0,2,2,"001234500009|ABCDEFGF"
PRINT 1
```

- 打印输出
- 示例代码8

```
SIZE 4,1
GAP 0,0
DIRECTION 1
CLS
REM UPCE Rule 4: 00abcde0000j = abcdej where j = 5-9
RSS 10,10,"UPCE",0,2,2,"001234500005|ABCDEFGF"
RSS 160,10,"UPCE",0,2,2,"001234500006|ABCDEFGF"
RSS 310,10,"UPCE",0,2,2,"001234500007|ABCDEFGF"
RSS 460,10,"UPCE",0,2,2,"001234500008|ABCDEFGF"
RSS 610,10,"UPCE",0,2,2,"001234500009|ABCDEFGF"
PRINT 1
```

- 打印输出
- 示例代码9

```
SIZE 4,1
GAP 0,0
DIRECTION 1
CLS
REM UPCE Rule 4: 00abcde0000j = abcdej where j = 5-9
RSS 10,10,"UPCE",0,2,2,"001234500005|ABCDEFGF"
RSS 160,10,"UPCE",0,2,2,"001234500006|ABCDEFGF"
RSS 310,10,"UPCE",0,2,2,"001234500007|ABCDEFGF"
RSS 460,10,"UPCE",0,2,2,"001234500008|ABCDEFGF"
RSS 610,10,"UPCE",0,2,2,"001234500009|ABCDEFGF"
PRINT 1
```

- 打印输出
- 示例代码10

```
SIZE 4,2
GAP 0,0
DIRECTION 1
CLS
RSS 10,10,"UPCA",0,1,1,"12345678901|ABCDEFGF"
RSS 210,10,"UPCA",0,2,1,"12345678901|ABCDEFGF"
RSS 410,10,"UPCA",0,3,1,"12345678901|ABCDEFGF"
PRINT 1
```

- 打印输出
- 示例代码11

```
SIZE 4,2
GAP 0,0
DIRECTION 1
CLS
RSS 10,10,"UPCE",0,1,1,"001200000456|ABCDEFGF"
RSS 210,10,"UPCE",0,2,1,"001210000456|ABCDEFGF"
RSS 410,10,"UPCE",0,3,1,"001220000456|ABCDEFGF"
PRINT 1
```

- 打印输出

BITMAP

打印位图（BMP位图按位取反打印）。

语法：

BITMAP x,y,width,height,mode,bitmap data...

参数	描述
x	x 坐标
y	y 坐标
width	图片宽度（单位：字节）
height	图片高度（单位：点）
mode	图片模式： 0: 覆盖 1: 或 2: 异或
bitmap data	位图数据

示例：

- 数据说明

Y size 16dots	X size 2 bytes 1-byte								2-byte							
	0	1	2	3	4	5	6	7	0	1	2	3	4	5	6	7
1	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
2	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
3	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
4	0	0	0	0	0	1	1	1	1	1	1	1	1	1	1	1
5	0	0	0	0	0	0	1	1	1	1	1	1	1	1	1	1
6	0	0	0	1	0	0	0	1	1	1	1	1	1	1	1	1
7	0	0	0	1	1	0	0	0	1	1	1	1	1	1	1	1
8	0	0	0	1	1	1	0	0	0	1	1	1	1	1	1	1
9	0	0	0	1	1	1	1	0	0	0	1	1	1	1	1	1
10	0	0	0	1	1	1	1	1	0	0	0	1	1	1	1	1
11	0	0	0	1	1	1	1	1	1	0	0	0	1	1	1	1
12	0	0	0	1	1	1	1	1	1	1	0	0	0	1	1	1
13	0	0	0	1	1	1	1	1	1	1	1	0	0	0	1	1
14	0	0	0	1	1	1	1	1	1	1	1	1	0	1	1	1
15	0	0	0	1	1	1	1	1	1	1	1	1	1	1	1	1
16	0	0	0	1	1	1	1	1	1	1	1	1	1	1	1	1

- 示例代码

ASCII:	十六进制:
SIZE 4,2	53 49 5A 45 20 34 2C 32 0D 0A 47
GAP 0,0	41 50 20 30 2C 30 0D 0A 43 4C 53
CLS	0D 0A 42 49 54 4D 41 50 20 32 30
BITMAP 200,200,2,16,0,	30 2C 32 30 30 2C 32 2C 31 36 2C
?????	30 2C 00 00 00 00 00 07 FF 03
?????	FF 11 FF 18 FF 1C 7F 1E 3F 1F 1F
?????	1F 8F 1F C7 1F E3 1F E7 1F FF 1F
PRINT 1,1	FF 0D 0A 50 52 49 4E 54 20 31 2C
	31 0D 0A

- 打印输出



*相关指令: PUTBMP,PUTPCX

PUTBMP

打印 BMP 格式位图, 仅支持单色位图。

语法:

PUTBMP x,y,"filename"[,bpp][,contrast]

参数	描述
x	x 坐标
y	y 坐标
filename	下载的 BMP 文件名
bpp	可选项。每像素点位数 1: 1位, 黑白位图 8: 8位, 256色位图 (暂不支持)
contrast	可选项。对比度, 默认80。(暂不支持)

示例:

- 示例代码

```
SPEED 2
DENSITY 3
SIZE 4,1.5
GAP 0,0
DIRECTION 1
CLS
PUTBMP 10,10,"SAMPLE.BMP"
BLOCK 10,180,240,100,"2",0,1,1,"bpp and contrast are omitted."
PUTBMP 300,10, "SAMPLE.BMP",1,80
BLOCK 300,180,240,100,"2",0,1,1, "bpp = 1 contrast = 80"
PUTBMP 590,10, "SAMPLE.BMP",8,80
BLOCK 590,180,240,100,"2",0,1,1,"bpp = 8
contrast = 80"
PRINT 1
```

注：需先下载 "SAMPLE.BMP" 文件

- 相关指令：DOWNLOAD,BITMAP,PUTPCX

PUTPCX

打印 PCX 格式图片，仅支持2-色 PCX图片。

语法：

```
PUTPCX x,y,"filename"
```

参数	描述
x	x 坐标
y	y 坐标
filename	PCX 文件名

示例：

- 示例代码

```
SPEED 2
DENSITY 3
SIZE 4,1.5
GAP 0,0
DIRECTION 1
CLS
PUTBMP 10,10, "SAMPLE.PCX"
PRINT 1
```

注：需先下载 "SAMPLE.PCX" 文件

- 相关指令：DOWNLOAD,BITMAP,PUTPCX

BOX

绘制方框。

语法:

BOX x,y,x_end,y_end,line thickness[,radius]

参数	描述
x	左上角位置的 x 坐标 (单位: 点)
y	左上角位置的 y 坐标 (单位: 点)
x_end	右下角位置的 x 坐标 (单位: 点)
y_end	右下角位置的 y 坐标 (单位: 点)
line thickness	线宽 (单位: 点)
radius	可选项。圆角半径, 默认为0


示例:

- 示例代码

```

SIZE 4,1.1
CLS
BOX 60,60,610,210,4
BOX 80,80,590,190,4
BOX 100,100,570,170,4,20
BOX 120,120,550,150,4,20
PRINT 1
  
```

- 打印输出



- 相关指令: BAR

CIRCLE

绘制圆形。

语法:

CIRCLE x_start,y_start,diameter,thickness

参数	描述
x_start	左上角 x 坐标 (单位: 点)
y_start	左上角 y 坐标 (单位: 点)
diameter	直径 (单位: 点)
thickness	线宽 (单位: 点)


示例:

• 示例代码

```

SIZE 80 mm,30 mm
GAP 0,0
DIRECTION 1
CLS
BAR 250,20,100,1
BAR 250,20,1,100
CIRCLE 250,20,100,5
PRINT 1
    
```

• 打印输出



ELLIPSE

绘制椭圆。

语法:

ELLIPSE x,y,width,height,thickness

参数	描述
x	左上角 x 坐标 (单位: 点)
y	左上角 y 坐标 (单位: 点)
width	椭圆宽度 (单位: 点)
height	椭圆高度 (单位: 点)
thickness	线宽 (单位: 点)


示例:

• 示例代码

```

SIZE 4,3
GAP 0,0
DIRECTION 1
CLS
BOX 10,10,410,110,1
ELLIPSE 10,10,400,100,2
BOX 10,120,110,520,1
ELLIPSE 10,120,100,400,5
PRINT 1
    
```

- 打印输出



ERASE

清除指定区域。

语法:

```
ERASE x,y,x_width,y_height
```

参数	说明
x	起始点 x 坐标 (单位: 点)
y	起始点 y 坐标 (单位: 点)
x_width	区域宽度 (单位: 点)
y_height	区域高度 (单位: 点)

示例:

- 示例代码

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
BAR 100,100,300,300  
ERASE 150,150,200,200  
PRINT 1,1
```

- 打印输出



- 相关指令：CLS

REVERSE

反白指定的区域。

语法：

```
REVERSE x_start,y_start,x_width,y_height
```

参数	描述
x_start	起始位置 x 坐标
y_start	起始位置 y 坐标
x_width	x 轴方向宽度（单位：点）
y_height	y 轴方向高度（单位：点）

示例：

- 示例代码

```
SIZE 4,2.5  
GAP 0,0  
DIRECTION 1  
CLS  
TEXT 100,100,"3",0,1,1,"REVERSE"  
REVERSE 90,90,128,40  
PRINT 1,1
```

- 打印输出



DIAGONAL

绘制斜线。

语法:**DIAGONAL x1,y1,x2,y2,thickness**

参数	描述
x1	起点 x1 坐标 (单位: 点)
y1	起点 y1 坐标 (单位: 点)
x2	终点 x2 坐标 (单位: 点)
y2	终点 y2 坐标 (单位: 点)
thickness	线宽 (单位: 点)

示例:

• 示例代码

```
SIZE 4,2.5
GAP 0,0
DIRECTION 1
CLS
DIAGONAL 50, 200, 200, 50, 16
DIAGONAL 50, 500, 500, 50, 8
PRINT 1,1
```

• 打印输出

TEXT

打印文本。

语法:**TEXT x,y,"font",rotation,x-multi,y-multi,[alignment,]"content"**

参数	描述
x	文本打印的 x 坐标
y	文本打印的 y 坐标
font	字体名称
	0 矢量字体（是否支持参考具体机型）
	1 8 × 12 点阵字体
	2 12 × 20 点阵字体
	3 16 × 24 点阵字体
	4 24 × 32 点阵字体
	5 32 × 48 点阵字体
	6 14 × 19 点阵字体
	7 21 × 27 点阵字体
	8 14 × 25 点阵字体
rotate	文本旋转方向
	0: 不旋转
	90: 旋转90°
	180: 旋转180°
	270: 旋转270°
x-multi	水平方向放大倍数，范围：1~10
	对于font0和其它矢量字体，该参数指定字符宽度
y-multi	垂直方向放大倍数，范围：1~10
	对于font0和其它矢量字体，该参数指定字符高度
alignment	文本对齐方式
	0: 默认（左对齐）
	1: 左对齐
	2: 居中
	3: 右对齐
content	文本内容

- 注意：
 - 如果文本中含有（"），双引号需按如下格式发送：\"

示例：

- 示例代码

```

SIZE 3,3
GAP 0,0
DIRECTION 1
CLS
TEXT 10,10,"0",0,12,12,"TSPL 2"
BAR 0,70,800,4
TEXT 10,110,"0",0,12,12,"FONT 0"
TEXT 10,160,"1",0,1,1,"FONT 1"
TEXT 10,210,"2",0,1,1,"FONT 2"
TEXT 10,260,"3",0,1,1,0,"FONT 3"
TEXT 10,310,"4",0,1,1,0,"FONT 4"
TEXT 10,360,"5",0,1,1,0,"FONT 5"
TEXT 10,410,"6",0,1,1,1,"FONT 6"
TEXT 10,460,"7",0,1,1,1,"FONT 7"
TEXT 10,510,"8",0,1,1,1,"FONT 8"
PRINT 1
    
```

- 打印输出

TSPL 2

FONT 0

FONT 1

FONT 2

FONT 3

FONT 4

FONT 5

FONT 6

F O N T 7

F O N T 8


- 示例代码

```
SIZE 3,2
GAP 0,0
DIRECTION 1
CLS
BAR 60,120,200,1
BAR 160,20,1,200
TEXT 160,120,"0",0,12,12,1,"align left"
PRINT 1,1

SIZE 3,2
GAP 0,0
DIRECTION 1
CLS
BAR 60,120,200,1
BAR 160,20,1,200
TEXT 160,120,"0",0,12,12,2,"align center"
PRINT 1,1

SIZE 3,2
GAP 0,0
DIRECTION 1
CLS
BAR 60,120,200,1
BAR 160,20,1,200
TEXT 160,120,"0",0,12,12,3,"align right"
PRINT 1,1
```

- 打印输出



BLOCK

打印一个段落。

语法:

```
BLOCK x,y,width,height,"font",rotation,x-multi,y-multi,[space,]align,]fit,]"content"
```

参数	描述
x	文本打印的 x 坐标
y	文本打印的 y 坐标
width	打印区域宽度（单位：点）
height	打印区域高度（单位：点）
font	字体名称
	0 矢量字体（是否支持参考具体机型）
	1 8 × 12 点阵字体
	2 12 × 20 点阵字体
	3 16 × 24 点阵字体
	4 24 × 32 点阵字体
	5 32 × 48 点阵字体
	6 14 × 19 点阵字体
	7 21 × 27 点阵字体
	8 14 × 25 点阵字体
rotate	文本旋转方向
	0: 不旋转
	90: 旋转90°
	180: 旋转180°
	270: 旋转270°
x-multi	水平方向放大倍数，范围：1~10
	对于font0和其它矢量字体，该参数指定字符宽度
y-multi	垂直方向放大倍数，范围：1~10
	对于font0和其它矢量字体，该参数指定字符高度
[space]	增加或删除行间空白（单位：点）
[align]	文本对齐方式
	0: 默认（左对齐）
	1: 左对齐
	2: 居中
	3: 右对齐
[fit]	缩小文本以适应打印区域大小（暂不支持）
content	文本内容

- 注意：
 - 如果文本中含有 (")，双引号需按如下格式发送：\"
 - \R代表回车符 0x0D
 - \L代表换行符 0x0A

示例：

- 示例代码


```

SIZE 3,0.5
GAP 0,0
DIRECTION 0
CLS
BOX 0,10,576,100,2
BLOCK 4,15,570,90,"1",0,1,1,"We stand behind our products with one of the most comprehensive
support programs in the Auto-ID industry."
PRINT 1
CLS
BOX 0,10,576,100,2
BLOCK 4,15,570,90,"1",0,1,1,20,2,"We stand behind our products with one of the most comprehe
nsive support programs in the Auto-ID industry."
PRINT 1
    
```

- 打印输出

```

We stand behind our products with one of the most comprehensive
support programs in the Auto-ID industry.
    
```

```

We stand behind our products with one of the most comprehensive
support programs in the Auto-ID industry.
    
```

状态查询及实时指令

<ESC>!?

该指令可以在任何时候获取打印机状态，即使是在打印机错误状态下。通过发送一个 <ESC> (ASCII 27) 作为起始控制字符请求查询状态。打印机返回一个字节，标志打印机状态。0表示打印机就绪。

语法：

```
<ESC>!?
```

返回状态说明：

十六进制	打印机状态
00	打印机就绪
01	开盖
02	卡纸
03	卡纸, 开盖

十六进制	打印机状态
04	缺纸
05	缺纸, 开盖
08	缺碳带
09	缺碳带, 开盖
0A	缺碳带, 卡纸
0B	缺碳带, 卡纸, 开盖
0C	缺碳带, 缺纸
0D	缺碳带, 缺纸, 开盖
10	暂停
20	正在打印
80	其它错误

- 相关指令: <ESC>IS

<ESC>!C

重启打印机并删除 AUTO.BAS。命令起始字符为<ESC> (ASCII 27)。

语法:

```
<ESC>!C
```

- 相关命令: <ESC>!Q

<ESC>!D

禁止实时指令, 例如: <ESC>!R, <ESC>!?, <ESC>!C 等以 <ESC>! 开头的实时指令。命令起始字符为<ESC> (ASCII 27)。

语法:

```
<ESC>!D
```

- 相关命令: ~!E

<ESC>!O

取消暂停模式。命令起始字符为<ESC> (ASCII 27)。

语法:

```
<ESC>!O
```

- 相关指令: <ESC>!P

<ESC>!P

设置暂停模式。命令起始字符为<ESC> (ASCII 27)。

语法:

```
<ESC>!P
```

- 相关指令: <ESC>!O

<ESC>!Q

重启打印机并删除 AUTO.BAS。命令起始字符为<ESC> (ASCII 27)。

语法:

```
<ESC>!Q
```

- 相关命令: <ESC>!C

<ESC>!R

重置打印机。命令起始字符为<ESC> (ASCII 27)。重置后打印机内下载文件将被删除, 该指令不能再DUMP模式使用。

语法:

```
<ESC>!R
```

- 相关命令: <ESC>!?

<ESC>!S

该指令可以在任何时候获取打印机状态, 即使是在打印机错误状态下。通过发送一个 <ESC> (ASCII 27) 作为起始控制字符请求查询状态。打印机返回 8 个字节, 标志打印机状态。

语法:

```
<ESC>!S
```

回复数据格式:

```
<STX>[4 字节状态]<ETX><CR><LF>
```

• 第一个字节

Bit7	Bit6	Bit5	Bit4	Bit3	Bit2	Bit1	Bit0	Hex	ASCII	Char	说明
0	1	0	0	0	0	0	0	40	64	@	正常
0	1	1	0	0	0	0	0	60	96	`	暂停
0	1	0	0	0	0	1	0	42	66	B	正在回退标签
0	1	0	0	0	0	1	1	43	67	C	正在切纸
0	1	0	0	0	1	0	1	45	69	E	打印机错误
0	1	0	0	0	1	1	0	46	70	F	定位
0	1	0	0	1	0	1	1	4B	75	K	等待按下打印按键
0	1	0	0	1	1	0	0	4C	76	L	等待取走标签
0	1	0	1	0	0	0	0	50	80	P	正在批量打印
0	1	0	1	0	1	1	1	57	87	W	正在绘图

• 第二个字节

Bit7	Bit6	Bit5	Bit4	Bit3	Bit2	Bit1	Bit0	Hex	ASCII	Char	说明
0	1	0	0	0	0	0	0	40	64	@	正常
0	1	0	0	0	0	0	1	41	65	A	缺纸
0	1	0	0	0	0	1	0	42	66	B	缺碳带
0	1	0	0	0	1	0	0	44	68	D	保留
0	1	0	0	1	0	0	0	48	72	H	接收缓存满
0	1	1	0	0	0	0	0	60	96	`	保留

• 第三个字节

Bit7	Bit6	Bit5	Bit4	Bit3	Bit2	Bit1	Bit0	Hex	ASCII	Char	说明
0	1	0	0	0	0	0	0	40	64	@	正常
0	1	0	0	0	0	0	1	41	65	A	打印头过热
0	1	0	0	0	0	1	0	42	66	B	马达过热
0	1	0	0	0	1	0	0	44	68	D	打印头错误
0	1	0	0	1	0	0	0	48	72	H	卡纸

Bit7	Bit6	Bit5	Bit4	Bit3	Bit2	Bit1	Bit0	Hex	ASCII	Char	说明
0	1	1	0	0	0	0	0	60	96	`	内存不足

- 第四个字节

Bit7	Bit6	Bit5	Bit4	Bit3	Bit2	Bit1	Bit0	Hex	ASCII	Char	说明
0	1	0	0	0	0	0	0	40	64	@	正常
0	1	0	0	0	0	0	1	41	65	A	无纸
0	1	0	0	0	0	1	0	42	66	B	卡纸
0	1	0	0	0	1	0	0	44	68	D	无碳带
0	1	0	0	1	0	0	0	48	72	H	卡碳带
0	1	1	0	0	0	0	0	60	96	`	开盖

- 相关指令：<ESC>!F

<ESC>!.

取消所有打印文件。命令起始字符为<ESC> (ASCII 27)。

语法：

```
<ESC>!.
```

~!@

获取打印机里程计。以ASCII字符返回里程计的整数部分，以0x0D结尾。

语法：

```
~!@
```

~!A

获取打印机空闲内存大小。打印机以十进制返回，以0x0D结尾。

语法：

```
~!A
```

- 相关指令：FILES

~!C

获取打印机 RTC 状态，打印机返回一个字节指示是否安装 RTC。仅在支持 RTC 机型上使用。

语法：

```
~!C
```

- 返回值：
 - 0: 未安装 RTC
 - 1: 已安装 RTC

~!D

打印机进入 DUMP 模式，DUMP 模式下，打印机直接打印接收到的数据。

语法：

```
~!D
```

~!E

允许实时指令执行。例如 <ESC>!R <ESC>!? <ESC>!C 等以 <ESC>! 开头的指令。

语法：

```
~!E
```

~!F

获取打印机内置文件、字库信息，文件名以ASCII字符返回，每个文件名以 0x0D 结尾，结束字符为 0x1A 。

语法：

```
~!F
```

- 相关指令：FILES

~!I

获取打印机代码页或国际字符集。

语法：

```
~!I
```

- 返回数据格式，例如：
 - 8位：431,001
 - 7位：USA,001
- 相关指令：COUNTRY, CODEPAGE

~!T

获取打印机名称。

语法:

```
~!T
```

文件管理命名

DOWNLOAD

文件下载。

语法:

- 下载程序文件

```
DOWNLOAD [n,]"FILENAME.BAS"
```

参数	描述
n	指定存储文件的内存区域 省略: 下载文件到 DRAM。 F: 下载文件到 flash E: 下载文件到扩展内存区域
FILENAME.BAS	文件名

- 注意:
 - 文件名区分大小写
 - 文件扩展名必须为“.BAS”
 - 文件名必须是 8.3 格式
 - 必须以 EOP 结尾
 - 如果没有指定内存区域, 默认下载到 DRAM 区域
 - DRAM 区域断电不保存
- 下载数据文件

```
DOWNLOAD [n,]"FILENAME",DATA SIZE,DATA CONTENT...
```

参数	描述
n	指定存储文件的内存区域 省略: 下载文件到 DRAM。 F: 下载文件到 flash E: 下载文件到扩展内存区域
FILENAME	文件名
DATA SIZE	数据长度
DATA CONTENT	数据内容

- 注意:
 - 对于文本数据, *CR* (回车) *0x0D* 和 *LF* (换行) *0x0A* 为数据分隔
 - 如果没有指定内存区域, 默认下载到 *DRAM* 区域
 - *DRAM* 区域断电不保存
 - 必须以 *EOP* 结尾

示例:

- 示例代码

```
DOWNLOAD "EXAMPLE.BAS"
SIZE 4,4
GAP 0,0
DIRECTION 1
SET TEAR ON
CLS
TEXT 100,100, "3",0,1,1, "EXAMPLE PROGRAM"
PRINT 1
EOP

DOWNLOAD F, "EXAMPLE.BAS"
SIZE 4,4
GAP 0,0
DIRECTION 1
SET TEAR ON
CLS
TEXT 100,100, "3",0,1,1, "EXAMPLE PROGRAM"
PRINT 1
EOP
```

- 相关指令: *EOP*,*RUN*,*PUTBMP*,*PUTPCX*,*INPUT*,*FILES*,*~!F*

EOP

结束进程。

语法:

EOP

- 相关指令: *DOWNLOAD*,*INPUT*,*FILES*,*~!F*

FILES

打印内存信息和文件列表。

语法:

FILES

- 相关指令: *~!F*,*KILL*

KILL

删除文件，通配符 (*) 指示删除所有文件。

语法:

```
KILL [n,]"FILENAME"
```

参数	描述
n	指定存储文件的内存区域 省略：下载文件到 DRAM。 F：下载文件到 flash E：下载文件到扩展内存区域
FILENAME	删除的文件名

- 注意：*
 - 如果未指定 *n*，默认删除 DRAM 中的文件
 - 用例：
 - KILL "FILENAME"：删除 DRAM 中指定的文件
 - KILL "*.PCX"：删除 DRAM 中所有 PCX 文件
 - KILL "*"：删除 DRAM 中所有文件
 - KILL F,"FILENAME"：删除 FLASH 中指定文件
 - KILL E,"*.PCX"：删除扩展内存中所有 PCX 文件
- 相关指令：